> -- tvorimvsevmeste писал(а):
> практикующие  педагоги, с которыми можно консультироваться?

опыт есть, а вот результатов...

> -- Цитата:
> Практически все приходящие в кружок имеют очень низкий уровень готовности напрягаться. Они все ожидают, что приложив минимум (нет, — микроскопическое количество) усилий, будут получать всевозможные награды и поощрения, как в играх для соц.сетей. Есть предложения, как решать эту проблему в применении к кружкусо свободным посещением?

аналогичная ситуация с кружком, решения пока не нашел.
еще можно с Виталием Валерьевичем ( vvp ) посоветоваться. Он как раз кружками занимается.


общался с человеком, который ответсвенный за олимпиады в области, он тоже ведет одну группу детей в "доме творчества", цель олимпиадное программирование, но народ расползается быстро. от 12 остается 4,а с малыми группами "не работается".

возможно помогут интересные проекты.
на форуме были темы:
viewtopic.php?p=39973#p39973
там целый список.

0) еще была тема про подход Дединского, но там без напряга не пойдет
viewtopic.php?f=7&t=4375

1) один из проектов реализовать довольно просто.
создать простой фоторедактор.
в теме
viewtopic.php?p=76194#p76194
есть код, позволяющий подгружать какртинки из *.bmp  (правда у меня цвета корявяться почему-то)
есть там и ссылки на другие способы.

если картинка представленна массивом, ее уже можно:
поворачивать, играться с цветовой гаммой, накладывать фильтры (маски размытия и резкости краев).
можно ставить задачу "удаления красных глаз" или др

может быть это будет интересно?

2) игры
> -- Цитата:
> Есть мнение, что создание компьютерных игрушек может сильно заинтересовать
дйствительно, на ББ некоторые игры создавать очень просто, но...
на форуме более опытные учителя уточняли, что детям чаще хочется создавать бегалки-стрелялки, а не шахматы-шашки-рензи

можно написать исполнителя-заготовку для реализации игр на клетчатом поле.
придумывал игры на исполнителе "паркетчик":
- игры без анимации(можно без нее): крестики-нолики (и подобные), морской бой, лайнс. в них можно реализовывать игру с компьютером, а для некоторых и турнир у кого лучше алгоритм. для реализации требуется процедура считывания клика мышки.
- игры с анимацией (тут она необходима): "жизнь", змейка, тетрис!, пакмен. тут требуется анимация, чтобы видеть процесс. необходима процедур считывания клавиш, и отложенные выполнения шагов игры.

если сделать поле заготовку с более широкими возможностями, чем "паркетчик" (добавить числа в клетки), то можно реализовать минера.

на паркетчике змейка выглядит вот так:
## Код:
```
МОДУЛЬ  Задание;
   ПОДКЛЮЧИТЬ  M := PrivMessenger, Ran := i21eduRandom, П := i21eduParketAnimRu, Ввод := i21eduВвод, Вывод := i21eduВывод;
   
   КОНСТАНТА
      вверх=1;
      вправо=2;
      вниз=3;
      влево=4;
      
   
   ПЕР
      направление : ЦЕЛ;
      старт : ЛОГИЧ;
      конец : ЛОГИЧ;
      
      змейка : МАССИВ 10,2 ИЗ ЦЕЛЫХ;
      длина : ЦЕЛ;
      
   ПРОЦЕДУРА ОбновиНаправление(номер : ЦЕЛ);
   НАЧАЛО
      ЕСЛИ НЕ старт ТО
         старт :=ДА;
      КОНЕЦ;
      
      ЕСЛИ номер=30 ТО
         направление:=вверх;
      АЕСЛИ номер=28 ТО
         направление:=влево;
      АЕСЛИ номер=31 ТО
         направление:=вниз;
      АЕСЛИ номер=29 ТО
         направление:=вправо;
      АЕСЛИ номер=27 ТО
         конец := ДА;
      КОНЕЦ;
   
   КОНЕЦ ОбновиНаправление;
   
   ПРОЦЕДУРА ПоложитьЕду;
      ПЕР
         положил : ЛОГИЧ;
         х,у : ЦЕЛ;
   НАЧАЛО
      положил := НЕТ;
      
      ПОКА НЕ положил ДЕЛАТЬ
         х:=Ran.Int(10)+1;
         у:=Ran.Int(10)+1;
         П.ПерейтиНа( х , у );
         ЕСЛИ НЕ (П.Красная() ИЛИ П.Зеленая()) ТО
            П.Положить( К );
            положил := ДА;
         КОНЕЦ;
      КОНЕЦ;
      П.ПерейтиНа( змейка[0,0] , змейка[0,1] );
   КОНЕЦ ПоложитьЕду;
   
   ПРОЦЕДУРА Ползи;
      ПЕР
         к,х,у : ЦЕЛ;
         измДлины : ЛОГИЧ;
   НАЧАЛО
            
      х := змейка[0,0];
      у := змейка[0,1];
      ЕСЛИ направление=вверх ТО
         у := у +1;
      АЕСЛИ направление=влево ТО
         х := х-1;
      АЕСЛИ направление=вниз ТО
         у := у-1;
      АЕСЛИ направление=вправо ТО
         х := х+1;
      КОНЕЦ;
      П.ПерейтиНа( х , у );
      
      измДлины := П.Красная();
      ЕСЛИ П.Красная() ТО
         П.СнятьПлитку;
         
         (*увеличиваем длину змеи*)
         ЕСЛИ длина < ДЛИНА(змейка,0)ТО
            длина := длина+1;
            змейка[длина-1,0]:=змейка[длина-2,0];
            змейка[длина-1,1]:=змейка[длина-2,1];
         ИНАЧЕ
            измДлины := НЕТ;
         КОНЕЦ;
         
         ПоложитьЕду;
         
      КОНЕЦ;
      
      ЕСЛИ НЕ измДлины ТО
         П.ПерейтиНа( змейка[длина-1,0] , змейка[длина-1,1] );
         П.СнятьПлитку;
      КОНЕЦ;
      
      к := длина - 1;
      ПОКА к>0 ДЕЛАТЬ
         змейка[к,0]:=змейка[к-1,0];
         змейка[к,1]:=змейка[к-1,1];
         к := к-1;
      КОНЕЦ;
      змейка[0,0]:=х;
      змейка[0,1]:=у;
      
      П.ПерейтиНа( х , у );
      П.Положить( З );
      
   КОНЕЦ Ползи;
   
   ПРОЦЕДУРА Ход;
   НАЧАЛО
      ЕСЛИ старт ТО
         Ползи;
      КОНЕЦ;
      
      ЕСЛИ НЕ конец ТО
         П.Отложить(Ход, 500);
      КОНЕЦ;
   КОНЕЦ Ход;

НАЧАЛО
   П.НаСтарт;
   
   змейка[0,0]:=1;
   змейка[0,1]:=5;
   длина := 1;
   П.ПерейтиНа( змейка[0,0] , змейка[0,1] );
   П.Положить( З );
   ПоложитьЕду;
   
   
   направление :=вправо;
   старт := ЛОЖЬ;
   конец := ЛОЖЬ;
   
   П.ПриНажатии(ОбновиНаправление);
   П.Отложить(Ход, 100);

КОНЕЦ Задание.
```



но вот я такими простыми играми заинтересовать детей не смог.

3) Александр Ильин в теме
viewtopic.php?p=61374#p61374
упоминал, что сделал поддержку скинов для черепашки. возможно у него есть другие примеры.


в целом, судя по обрывкам фраз с форума у людей есть опыт работы с детьми на ББ.
но в одном месте это не собрано, не представлено.
было бы здорово, если бы люди рассказали, чем занимаются/занимались с детьми.

4) во вложении описан алгоритм обучаейся игры в крестики-нолики.